Formal definitions, theorems, algorithm, and implementation

1 Introduction

In this document we present the definitions, results, and the algorithm that appear in our paper 'Continuous double auctions: characterization and formalization' alongside their formal versions (included in the shaded boxes), making it convenient for a reader to understand the formalization. These formal statements are taken directly from the accompanying Coq formalization. Also, we include the details of the application that results from this work.

2 Definitions

In this section, we introduce and define all the notions that help us establish our results. In the presentation of our work, we make extensive use of sets for ease of readability for a general mathematical audience. In the Coq formalization, we use lists instead. The presentation and the Coq formalization closely mirror each other and the mathematical content is kept intact. The choice of lists instead of sets in the formalization helps us in two major ways: we can use some of the existing publicly available libraries from previous works for the purposes of modelling auctions, and, more importantly, it helps us in extracting a reasonably fast OCaml program for the verified algorithm which we apply on real data generated from an exchange.

2.1 Orders

We model both bids and asks as orders. This is a major diversion from earlier works on formalizing call auctions. This allows us to streamline the formalization by avoiding proving the common properties of bids and asks twice. Below we define orders and associated functions.

Definition 1 (order, id, timestamp, qty, price, ids). An order is a 4-tuple (id, time, quantity, price) where each component is a natural number and quantity > 0. An order is used to represent a bid or an ask which has the attributes of id, timestamp, quantity and the limit price. We can extract the components of an order using the functions id, timestamp, qty, and price.

```
(* Definition of Order *)
Record order:Type:=
Mk_order{
   id: nat;
   otime: nat;
   oquantity: nat;
   oprice: nat;
   oquantity_cond: Nat.ltb oquantity 1 = false
}.
```

For a set of orders Ω , we define

```
\mathsf{ids}(\Omega) := \{id \mid \exists \omega \in \Omega \ \textit{s.t.} \ \mathsf{id}(\omega) = id\}.
```

In our formalization, $\mathsf{ids}(\Omega)$ allows for multiple copies of the same id (when Ω has multiple orders with the same id. Whenever we need the collection of distinct ids of Ω , we use $\mathsf{uniq}\ \mathsf{ids}(\Omega)$, where $\mathsf{uniq}\ \mathsf{returns}$ only one copy of each entry in a list.

```
(*Defintion of ids*)
Fixpoint ids (B:list order):(list nat):=
  match B with
  |nil => nil
  |b::B' => (id b)::ids B'
  end.
```

For a set of orders Ω with distinct ids and id which appears in Ω , with slight abuse of notation, we define $\operatorname{timestamp}(\Omega,id)$, $\operatorname{qty}(\Omega,id)$, and $\operatorname{price}(\Omega,id)$ to be $\operatorname{timestamp}(\omega)$, $\operatorname{qty}(\omega)$, and $\operatorname{price}(\omega)$ (respectively) of the order $\omega \in \Omega$ such that $\operatorname{id}(\omega) = id$. Below is a function to extract price of an order from a list of order for a given is. Similarly, we define $\operatorname{timestamp}$ and qty functions.

```
(*Definition of price(B, i)*)
(*Only used when In i (ids B)*)
Fixpoint price (B: list order)(i:nat):=
  match B with
  |nil => 0
  |b::B' => if (id b) == i then oprice b else price B' i end.
```

Similarly, we have defined timestamp and quantity functions in Coq.

We will often have a universe from which the bids and asks arise. To capture this notion, we define an order domain.

Definition 2 (order-domain, admissible). (B, A) is an order-domain, if B and A are sets of orders.

In an order-domain (B, A), the first component B represents a set of bids and the second component A represents a set of asks.

An order-domain is called admissible where each id is distinct and each timestamp is distinct.

We do not need to explicitly define order domain in Coq. We use the following instead.

```
(*(B,A) is an order domain*)
(B A : list order)
```

We use a slightly more general definition of admissible in Coq than what we present in the paper. Instead of requiring all ids (timestamps) to be distinct, we just require the ids (timestamps) of the bids to be distinct and the ids (timestamps) of the asks to be distinct. In the following NoDup (stands for 'No Duplicate') is a proposition which is True iff there are no duplicates.

```
Definition admissible (B A :list order) :=
(NoDup (ids B))/\(NoDup (ids A))/\
(NoDup (timesof B))/\(NoDup (timesof A)).
```

Once we have the notion of an order-domain, we do not need to define bids and asks explicitly; if an order belongs to the first component of an order-domain, it is regarded as a bid and if it belongs to the second component, it is regarded as an ask.

We now define when a bid and an ask are tradable and when an order-domain is matchable.

Definition 3 (tradable, matchable). Given two orders b (bid) and a (ask), we say b and a are tradable if $price(b) \ge price(a)$. Given an order-domain (B, A), B and A are matchable if there exists $b \in B$ and $a \in A$ such that b and a are tradable.

```
Definition tradable (b a: order):= (oprice b >= oprice a).

Definition matchable (B A : list order):= exists b a, (In a A)/\(In b B)/\(tradable b a).
```

Next, we capture the notion of competitiveness between two asks and between two bids based on price and time.

Definition 4 (Competitiveness, \succ). A bid b_1 is defined to be more competitive compared to another bid b_2 , denoted by $b_1 \succ b_2$, iff

```
\begin{array}{l} \operatorname{price}(b_1) > \operatorname{price}(b_2) \ OR \ (\operatorname{price}(b_1) = \operatorname{price}(b_2) \ AND \\ \operatorname{timestamp}(b_1) < \operatorname{timestamp}(b_2)). \ Similarly, \ An \ ask \ a_1 \ is \ considered \ more \ \operatorname{competitive} \ compared \ to \ another \ ask \ a_2, \ denoted \ by \ a_1 \succ a_2, \ iff \\ \operatorname{price}(a_1) < \operatorname{price}(a_2) \ OR \ (\operatorname{price}(a_1) = \operatorname{price}(a_2) \ AND \\ \operatorname{timestamp}(a_1) < \operatorname{timestamp}(a_2)). \end{array}
```

In our Coq formalization, we use two definitions for the competitive operator, namely becompetitive and acompetitive, one for comparing bids and one for comparing asks.

```
Definition bcompetitive (b b':order):=
(Nat.ltb (oprice b') (oprice b)) ||
((Nat.eqb (oprice b') (oprice b)) &&
(Nat.leb (otime b) (otime b'))).

Definition acompetitive (a a':order):=
(Nat.ltb (oprice a) (oprice a')) ||
((Nat.eqb (oprice a) (oprice a')) &&
(Nat.leb (otime a) (otime a'))).
```

We treat a set of orders also as a multiset where we suppress the quantity field of each order and set its multiplicity equal to its quantity. This view will help us succinctly state the conservation property and ease the formalization substantially. For two sets of orders R and S, we define the relation R-S, which corresponds to the usual notion of set difference when viewing R and S as multisets. We continue to use the usual \backslash when the sets are not treated as multisets.

Definition 5 (– between sets of orders). Let Ω_1 and Ω_2 be sets of orders such that each of them contains orders with distinct ids.

```
\begin{split} \Omega_1 - \Omega_2 := \\ \{(id,t,q,p) \mid \exists (id,t,q_1,p) \in \Omega_1 \ and \ \exists (id,t,q_2,p) \in \Omega_2 \ s.t. \\ q = q_1 - q_2 \ and \ q > 0 \\ OR \\ \exists ((id,t,q,p) \in \Omega_1 \ and \ \forall q'(id,t,q',p) \notin \Omega_2\}. \end{split}
```

2.2 Transactions and matchings

Here, we define transactions, matchings, and associated functions.

In a major modelling decision, we opted to not keep price and timestamp in a transaction which are usually kept in real data for many applications. This greatly simplifies our formalization and exposition without diluting any mathematical content. Based on the application, a transaction between tradable orders b and a can be assigned an appropriate transaction price in the interval $[\operatorname{price}(a),\operatorname{price}(b)]$. In stock markets for example, this price may vary from exchange to exchange. For our work though the role of price and timestamp is used to only decide competitiveness and checking tradability between orders; keeping the price or timestamp in a transaction is completely redundant.

We first define transactions and associated functions.

Definition 6 (transaction, id_{bid} , id_{ask} , qty, ids_{bid} , ids_{ask} , Qty, Vol). A transaction is a 3-tuple (id_b , id_a , quantity) of natural numbers where quantity > 0. Here id_b and id_a represent the ids of the participating bid and ask.

We can extract the components of a transaction using the functions id_{bid} , id_{ask} , and qty. Note that qty is overloaded, but that is for ease of presentation; formally they are kept different.

Given a set of transactions T, id_b and id_a , we define $\mathsf{ids}_{\mathsf{bid}}(T)$ and $\mathsf{ids}_{\mathsf{ask}}(T)$ as follows.

$$\mathsf{ids}_{\mathsf{bid}}(T) := \{id_b \mid \exists t \in T \ s.t. \ \mathsf{id}_{\mathsf{bid}}(t) = id_b\}.$$
$$\mathsf{ids}_{\mathsf{ask}}(T) := \{id_a \mid \exists t \in T \ s.t. \ \mathsf{id}_{\mathsf{ask}}(t) = id_a\}.$$

Given a set of transactions T, id_b and id_a we define $\mathsf{Qty}_{bid}(T,id_b)$ and $\mathsf{Qty}_{ask}(T,id_a)$ as follows.

$$\mathsf{Qty}_{bid}(T,id_b) := \sum_{t \in T: \ \mathsf{id}_{\mathsf{bid}}(t) = id_b} \mathsf{qty}(t),$$

which represents the total traded quantity of id_b in T. Similarly, we define the total traded quantity of id_a in T:

$$\operatorname{Qty}_{ask}(T,id_a) := \sum_{t \in T : \ \operatorname{id}_{\operatorname{ask}}(t) = id_a} \operatorname{qty}(t).$$

For ease of readability, whenever it is clear from context, we will use just Qty instead of Qty_{ask} and Qty_{bid} .

Given a set of transactions T, we define the total volume of T, denoted by Vol(T), as the sum of the quantities of the transactions in T. Formally,

$$\mathsf{Vol}(T) := \sum_{t \in T} \mathsf{qty}(t).$$

Often in Coq we keep two definitions for the same object, one is constructive whereas the other is propositional. We formally establish that these definitions are equivalent, and depending on the context one may be more useful than the other.

```
(* Definitions of ids_bid *)
Fixpoint ids_bid_aux (T: list transaction):(list nat):=
   match T with
   |nil => nil
   |t1::T' => (idb t1)::(ids_bid_aux T')
   end.

Definition fun_ids_bid (T:list transaction) :=
   (uniq (ids_bid_aux T)).

Definition ids_bid (I:list nat)(T:list transaction):=
   (forall i, In i I ->(exists t, (In t T)/\(idb t = i))) /\( (forall t, In t T -> (exists i, (In i I)/\(idb t = i))) /\( (NoDup I).
```

```
(*Definition of Vol*)
Fixpoint Vol (T: list transaction):(nat):=
  match T with
  |nil => 0
  |t::T' => tquantity t + (Vol T')
  end.
```

Next, we capture the notion of when a transaction can arise from an order-domain.

Definition 7 (over, valid). We say a transaction t is over the order-domain (B,A) iff $\mathsf{id}_{\mathsf{bid}}(t) = \mathsf{id}(b)$ for some $b \in B$ and $\mathsf{id}_{\mathsf{ask}}(t) = \mathsf{id}(a)$ for some $a \in A$. We say that a transaction t is valid w.r.t order-domain (B,A) iff there exists $b \in B$ and $a \in A$ such that

- $id_{bid}(t) = id(b)$ and $id_{ask}(t) = id(a)$.
- $price(a) \le price(b)$ (i.e., b and a are tradable).
- $qty(t) \le \min(qty(b), qty(a))$.

We say that a set of transactions T is valid over an order-domain (B, A) if each transaction in T is valid over $(B, A)^1$.

¹In our main paper, we just define valid for admissible domains, that simplifies the definition, and which is what we eventually use in the results.

```
Definition over (t : transaction)(B A : list order):=
exists b a, (In a A)/\(In b B)/\(idb t = id b)/\(ida t = id a).

Definition valid (t : transaction)(B A : list order):=
exists b a, (In a A)/\(In b B)/\(idb t = id b)/\(ida t = id a)/\(tradable b a)/\(tquantity t <= oquantity b)/\(tquantity t <= oquantity a).

Definition Tvalid (T : list transaction)(B A : list order):=
forall t, (In t T) -> (valid t B A).
```

Given a set of transactions, we would like to extract out the partial bids or the asks that got traded. Observe that to extract out these orders, we would need to determine the price and the timestamp of those orders, which is not available in the transactions; these sets can be determined if we know the underlying order-domain which gives rise to the transactions. Thus, we define the functions Bids and Asks as follows.

Definition 8 (Bids, Asks). Given a set of transactions T over an admissible order-domain (B,A) we define Bids(T,B) and Asks(T,A) as follows. Bids(T,B) is the set of all the bids participating in T where the quantity of a bid b is set to the total traded quantity of b in T. Formally,

```
\begin{split} \operatorname{Bids}(T,B) := \{(id,\operatorname{timestamp}(B,id),\operatorname{Qty}_{bid}(T,id),\operatorname{price}(B,id))\mid id\in\operatorname{ids}_{\operatorname{bid}}(T)\}.\\ Similarly, \end{split}
```

$$\mathsf{Asks}(T,A) := \{(id,\mathsf{timestamp}(A,id),\mathsf{Qty}_{ask}(T,id),\mathsf{price}(A,id)) \mid id \in \mathsf{ids}_{\mathsf{ask}}(T)\}.$$

We often simply write $\mathsf{Bids}(T)$ and $\mathsf{Asks}(T)$ instead of $\mathsf{Bids}(T,B)$ and $\mathsf{Asks}(T,A)$ whenever B and A are clear from the context.

```
(*Definitions of Bids*)
Fixpoint bids_aux (T: list transaction)(B:list order)
(Bi :list nat):(list order).
refine ( match Bi with
  |nil => nil
  |i::Bi' => match (Compare_dec.lt_dec (Qty_bid T i) 1) with
    |left _ => bids_aux T B Bi'
    |right _ => (Mk_order
                i (timestamp B i) (Qty_bid T i ) (price B i) _ )
                :: (bids_aux T B Bi')
 end
 end). rewrite PeanoNat.Nat.ltb_nlt. auto.
Defined.
Definition bids (T: list transaction)(B:list order):=
uniq (bids_aux T B (ids B)).
Definition Bids
(B:list order)(T: list transaction)(B': list order):=
subset (ids_bid_aux T) (ids B') ->
(forall b, In b B \rightarrow (exists t, (In t T)/\(idb t = id b)/\
(oquantity b = Qty_bid T (id b))/\
(exists b', (In b' B')/\(id b = id b')/\
(otime b = otime b')/\(oprice b = oprice b'))))/\
(subset (ids_bid_aux T) (ids B))/\(NoDup B).
```

Finally, we define a matching, which is a set of transactions that can simultaneously arise from an order-domain. We also define the canonical form of a set of transactions, which will be often applied to matchings.

Definition 9 (Matching, Canonical form). We say a set of valid transactions M over an order-domain (B,A) is a matching over (B,A) iff $\forall b \in B$, $\mathsf{qty}(b) \ge \mathsf{Qty}(M,\mathsf{id}(b))$ and $\forall a \in A$, $\mathsf{qty}(a) \ge \mathsf{Qty}(M,\mathsf{id}(a))$.

We define the canonical form of a set of transactions M, denoted by C(M), as follows.

$$\mathcal{C}(M) := \left\{ (id_b, id_a, q) \mid \exists m \in M \ s.t. \ \mathrm{id_{bid}}(m) = id_b, \mathrm{id_{ask}}(m) = id_b, \mathrm{id_{ask}}(m) = id_b, \mathrm{id_{bid}}(m) = id_b, \mathrm{id_{ask}}(m) =$$

Note that in a canonical form, for each participating bid and ask pair, there is a unique transaction.

```
(*Definitions of canonical form*)
Fixpoint cform_aux (T: list transaction)(Bi Ai :list nat):
(list transaction).
refine ( match (Bi, Ai) with
              |(nil, _) => nil
              |(_, nil) => nil
              |(i::Bi', j::Ai') =>
                    match (Compare_dec.lt_dec (Qty T i j) 1) with
                            |left _ => cform_aux T Bi' Ai'
                            |right _ => (Mk_transaction i j (Qty T i j) _ )::
                                                                     (cform_aux T Bi' Ai')
   end). rewrite PeanoNat.Nat.ltb_nlt. auto.
   Defined.
Definition cform (M: list transaction):(list transaction) :=
uniq (cform_aux M (ids_bid_aux M) (ids_ask_aux M)).
Definition CanonicalForm (M M': list transaction):=
(***M is canForm of M'***)
(forall m, In m M \rightarrow (Qty M' (idb m) (ida m) = tquantity m)/\
(exists m', (In m' M')/\(idb m = idb m')/\(ida m = ida m')))/\
(forall m', In m' M' ->
exists m, (In m M)/(idb m' = idb m)/(ida m' = ida m)/(ida m)/(ida m' = ida m)/(ida m)/(ida m)/(ida m' = ida m)/(ida 
(Qty M' (idb m) (ida m) = tquantity m)) /\(NoDup M).
```

2.3 Order-book and process

We now formally define instructions and order-book.

Definition 10 (command, instruction, order-book). Buy, Sell and Del are called commands.

An instruction is a pair (Δ, ω) where Δ is a command and ω is an order. For convenience we represent (Δ, ω) instruction by Δ ω . Also sometimes we represent (Del, ω) instruction simply by Del id (ω) ; this is done because for a Del command, only the id of the order matters.

An order-book is a list where each entry is an instruction.

```
(*Definition of command*)
Inductive command:Set:=
  |buy
  |sell
  |del.
```

```
(*Definition of instruction*)
Record instruction : Type:=
Mk_instruction { cmd : command; ord : order}.
```

We need to impose two technical conditions that an order-book should meet. Firstly, the timestamps of the orders must be increasing. Secondly, the ids of the orders in the non-Del instructions in the order-book must be all distinct. We will relax the second condition slightly that will help us in certain applications. We allow an order to have an id identical only to the id appearing in an immediately preceding Del order instruction. This will later help us in our application to implement an 'update' instruction, by replacing it with a delete instruction followed with a Buy or Sell instruction carrying the same id. We now formally define a 'structured' order-book that formally captures these conditions.

Definition 11 (structured order-book). An order-book $\mathcal{I} = [(\Delta_0, \omega_0), \cdots, (\Delta_n, \omega_n)]$ is called structured if the following conditions hold.

- For all $i \in \{0, \dots, n-1\}$, timestamp $(\omega_i) < \operatorname{timestamp}(\omega_{i+1})$.
- For all $i \in \{0, 1, \dots, n\}$, at least one of the following three conditions hold.
 - $\begin{array}{l} \text{-}\ \Delta_i = \mathsf{Del}.\\ \\ \text{-}\ \mathsf{id}(\omega_i) \notin \mathsf{ids}\{\omega_0,\cdots,\omega_{i-1}\}.\\ \\ \text{-}\ \mathsf{id}(\omega_i) = \mathsf{id}(\omega_{i-1})\ and\ \Delta_{i-1} = \mathsf{Del}. \end{array}$

```
Definition structured (I :list instruction):=
  (forall t:nat, (t+1) <= (length I) ->
  (cmd (nth t I tau0) = del)\/
  (~In (id (ord (nth t I tau0))) (ids (tilln (orders I) (t-1))))\/
  ((id (ord (nth t I tau0))) = (id (ord (nth (t-1) I tau0)))/\/
  (cmd (nth (t-1) I tau0) = del)))
  /\ Sorted (Nat.ltb) (timesof (orders I))/\/
  NoDup ((timesof (orders I))).
```

We now define a process which represents an abstract online algorithm that will be fed resident bids and asks and an instruction, and it will output a set of transactions and resulting resident bids and asks.

Definition 12 (process). A process is a function $(B, A, \tau) \mapsto (B', A', M)$ that takes as input sets of orders B, A and an instruction τ and outputs sets of orders B', A' and a set of transactions M.

We do not need to explicitly define process in Coq. It can be simply described by its type.

```
(*P is a process*)
(P :(list order)->(list order) -> instruction ->
(list order)*(list order)*(list transaction))
```

The input to a process is an order-domain, which represents the resident orders in the system, and an instruction. If this instruction, is a delete id instruction, then process is supposed to delete all resident orders with that id, and the 'effective' order-domain potentially gets reduced. Otherwise, if the instruction is a buy/sell order, then the 'effective' order-domain needs to include that order. We define Absorb that takes an order-domain and an instruction as input and outputs the 'effective' order-domain.

Definition 13 (Absorb). For an order-domain (B,A) and an instruction τ we define $\mathsf{Absorb}(B,A,\tau) :=$

$$\begin{cases} (\{\beta \in B | \mathsf{id}(\beta) \neq id\}, \{\alpha \in A | \mathsf{id}(\alpha) \neq id\}) \ \textit{if} \ \tau = \textit{Del} \ \textit{id} \\ (B \cup \{\beta\}, A) \ \textit{if} \ \tau = \textit{Buy} \ \beta \\ (B, A \cup \{\alpha\}) \ \textit{if} \ \tau = \textit{Sell} \ \alpha. \end{cases}$$

To a process, we will usually feed inputs that satisfy certain properties, and such inputs we refer to as legal-inputs and are defined as follows.

Definition 14 (legal-input). We say an order-domain (B, A) and an instruction τ forms a legal-input if B and A are not matchable and τ is such that $(B', A') = \mathsf{Absorb}(B, A, \tau)$ is an admissible order-domain.

```
Definition Legal_input (B A :list order)(tau: instruction):= admissible (fst (Absorb B A tau)) (snd (Absorb B A tau))/\
not (matchable B A).
```

Finally, we define Iterated.

Definition 15 (Iterated). Given a process P, an order-book $\mathcal I$ and a natural number k, we define $\operatorname{Iterated}(P,\mathcal I,k)$ to be the output of P at time k when it is iteratively run on the order-book $\mathcal I$. When $k > \operatorname{length}(\mathcal I)$, $\operatorname{Iterated}(P,\mathcal I,k)$ returns $(\emptyset,\emptyset,\emptyset)$. Otherwise, $\operatorname{Iterated}(P,\mathcal I,k)$ can be computed recursively as per the following algorithm.

```
Algorithm 1 Iteratively running a process on an order-book
```

```
function Iterated(Process P, Order-book \mathcal{I}, natural number k)

if k = 0 then return (\emptyset, \emptyset, \emptyset)

(B, A, M) \leftarrow \text{Iterated}(P, \mathcal{I}, k - 1)

\tau \leftarrow k^{\text{th}} instruction in \mathcal{I}
return P(B, A, \tau)
```

2.4 Three natural properties of a process

Having setup the definitions, we will now state the three natural properties formally.

We say a process P satisfies positive bid-ask spread, price-time priority, and conservation if for all order-domains (B,A) and an instruction τ such that (B,A) and τ forms a legal-input, $P(B,A,\tau)=(\hat{B},\hat{A},M)$ and $(B',A')=\mathsf{Absorb}(B,A,\tau)$ implies the following three conditions.

1. Positive Bid-Ask Spread: \hat{B} and \hat{A} are not matchable.

```
Definition Condition1 (M: list transaction)
(B A hat_B hat_A: list order) (tau: instruction):
Prop:= not (matchable hat_B hat_A).
```

2. Price-Time Priority: If a less competitive order ω is being traded in M, then all orders that are more competitive than ω must be fully traded in M. Formally,

```
a. \forall a, a' \in A', \ a \succ a' \text{ and } \mathsf{id}(a') \in \mathsf{ids}_{\mathsf{ask}}(M) \implies \mathsf{Qty}(M, id(a)) = \mathsf{qty}(a)
b. \forall b, b' \in B', \ b \succ b' \text{ and } \mathsf{id}(b') \in \mathsf{ids}_{\mathsf{bid}}(M) \implies \mathsf{Qty}(M, id(b)) = \mathsf{qty}(b).
```

```
Definition Condition2a (M: list transaction)(B:list order):Prop:=
forall b b', (In b B)/\(In b' B)/\((bcompetitive b b')/\((ln (id b') (ids_bid_aux M)) -> (Qty_bid M (id b)) = (oquantity b).
Definition Condition2b (M: list transaction)(A:list order):Prop:=
forall a a', (In a A)/\((ln a' A)/\((acompetitive a a'/\~eqcompetitive a a')/\((ln (id a') (ids_ask_aux M)) -> (Qty_ask M (id a)) = (oquantity a).
```

3. Conservation: P does not lose or add orders arbitrarily. For this we have the following technical conditions.

```
a. M is matching over the order-domain (B', A')
b. \hat{B} = B' - \mathsf{Bids}(M, B')
c. \hat{A} = A' - \mathsf{Asks}(M, A').
```

```
Definition Condition3a (M: list transaction)
(B A: list order) (tau: instruction):Prop:=
let B' := (fst (Absorb B A tau)) in
let A' := (snd (Absorb B A tau)) in
Matching M B' A'.

Definition Condition3b (M: list transaction)
(B A hat_B: list order) (tau: instruction):Prop:=
let B' := (fst (Absorb B A tau)) in
hat_B === (odiff B' (bids M B')).

Definition Condition3c (M: list transaction)
(B A hat_A: list order) (tau: instruction):Prop:=
let A' := (snd (Absorb B A tau)) in
(hat_A === (odiff A' (asks M A'))).
```

In Coq, we define the proposition Properties as follows. If a process P satisfies the above three conditions, then and only then 'Properties P' is True.

```
Definition Properties (Process: (list order) ->(list order) ->
instruction -> (list order)*(list order)*(list transaction)):=
forall A B tau, Legal_input B A tau ->
let B' := (fst (Absorb B A tau)) in
let A' := (snd (Absorb B A tau)) in
let hat_B := (Blist (Process B A tau)) in
let hat_A := (Alist (Process B A tau)) in
let M := (Mlist (Process B A tau)) in
Condition1 M B A hat_B hat_A tau /\
Condition2a M B'/\ Condition2b M A'/\
Condition3a M B A tau /\
Condition3b M B A hat_B tau /\
Condition3c M B A hat_A tau.
```

In the above we use Blist, Alist, and Mlist, which are defined as follows.

```
Definition Blist
(p: (list order)*(list order)*(list transaction)) :=
match p with (x, y,z) => x end.

Definition Alist
(p: (list order)*(list order)*(list transaction)) :=
match p with (x, y,z) => y end.

Definition Mlist
(p: (list order)*(list order)*(list transaction)) :=
match p with (x, y,z) => z end.
```

3 Verified algorithm

Here we introduce a natural algorithm for continuous double auctions.

In the Coq formalization of Process_instruction, we sort the list of asks and

bids by their competitiveness before calling a subroutine; as a result, the most competitive bid and ask are on top of their respective lists.

```
Definition Process_instruction (B A:list order)(tau: instruction):
  ((list order)*(list order)*(list transaction)):=
  match (cmd tau) with
    |del => Del_order B A (id (ord tau))
    |buy => Match_bid B (sort acompetitive A) (ord tau)
    |sell => Match_ask (sort bcompetitive B) A (ord tau)
end.
```

Algorithm 3 Matching an ask

```
function Match_ask(Bids B, Asks A, order \alpha)
                                                                                                                    \triangleright \alpha is an ask.
      if B = \emptyset then return (B, A \cup \{\alpha\}, \emptyset)
      \beta \leftarrow \mathsf{Extract\_most\_competitive}(B)
                                                                                                    \triangleright Note: B \leftarrow B \setminus \{\beta\}.
      if price(\beta) < price(\alpha) then return (B \cup \{\beta\}, A \cup \{\alpha\}, \emptyset)
                                                                        \triangleright From now on \beta and \alpha are tradable.
      if qty(\beta) = qty(\alpha) then
            m \leftarrow (\mathsf{id}(\beta), \mathsf{id}(\alpha), \mathsf{qty}(\alpha))
            return (B, A, \{m\})
      if qty(\beta) > qty(\alpha) then
            m \leftarrow (\mathsf{id}(\beta), \mathsf{id}(\alpha), \mathsf{qty}(\alpha))
             B' \leftarrow B \cup \{(\mathsf{id}(\beta), \mathsf{timestamp}(\beta), \mathsf{qty}(\beta) - \mathsf{qty}(\alpha), \mathsf{price}(\beta))\}
            return (B', A, \{m\})
      if qty(\beta) < qty(\alpha) then
             m \leftarrow (\mathsf{id}(\beta), \mathsf{id}(\alpha), \mathsf{qty}(\beta))
             \alpha' \leftarrow (\mathsf{id}(\alpha), \mathsf{timestamp}(\alpha), \mathsf{qty}(\alpha) - \mathsf{qty}(\beta), \mathsf{price}(\alpha))
            (B', A', M') \leftarrow \mathsf{Match\_ask}(B, A, \alpha')
            M \leftarrow M' \cup \{m\}
            return (B', A', M)
```

```
(*Definition of Match_ask(B,A,a)*)
Fixpoint Match_ask (B A: list order)(a :order):
((list order)*(list order)*(list transaction)).
destruct B as [|b B']. (*b is most competitive bid*)
exact (nil, a::A, nil). (*When B is empty*)
(*From now on B is not empty*)
destruct (oprice a - oprice b).
(*First case: price a <= price b, i.e., b & a tradable*)
(*destruct ((oquantity a) - (oquantity b)).*)
refine ( match
(Compare_dec.lt_eq_lt_dec (oquantity b) (oquantity a)) with
(*Subcase: qty b=qty a*)
 |inleft (right _) =>
 (B', A, ((Mk_transaction (id b) (id a) (oquantity a)
                           (oquantity_cond a))::nil))
(*Subcase qty b>qty a*)
 |inright _ =>
 ((Mk_order (id b) (otime b)
   ((oquantity b) - (oquantity a)) (oprice b) _)::B', A,
   ((Mk_transaction (id b) (id a) (oquantity a)
   (oquantity_cond a))::nil))
 (*Subcase: qty b < qty a*)
  |inleft (left _) =>
 let BAM := (Match_ask B' A (Mk_order (id a) (otime a)
  ((oquantity a) - (oquantity b)) (oprice a) _ )) in
  (Blist BAM, Alist BAM, (Mk_transaction (id b) (id a)
  (oquantity b) (oquantity_cond b))::(Mlist BAM))
 end ).
rewrite PeanoNat.Nat.ltb_nlt; apply liaforrun; auto.
rewrite PeanoNat.Nat.ltb_nlt; apply liaforrun;auto.
(*Second case: price a > price b, i.e., b & a not tradable*)
exact (b::B', a::A, nil). Defined.
```

The Match_Bid subroutine is symmetric to the Match_Ask subroutine and we do not present it explicitly here.

Algorithm 4 Deleting an order

```
Definition Del_order (B A:list order)(i:nat):
  ((list order)*(list order)*(list transaction)):=
  (delete_order B i, delete_order A i, nil).
```

4 Lemmas and Theorems

In the following Coq statements, 'A === B' for lists A and B represents that list A is a permutation of list B. If we think of A and B as sets, then 'A === B', translates to set equality A = B.

4.1 Maximum matching

Lemma 1 (Maximum matching). Let P be a process that satisfies **positive** bid-ask spread and conservation. For all order-domain and instruction pairs $((B,A),\tau)$ that form legal-inputs, if $P(B,A,\tau)=(\hat{B},\hat{A},M)$, then for all matchings M' over $\mathsf{Absorb}(B,A,\tau)$, $\mathsf{Vol}(M) \geq \mathsf{Vol}(M')$.

```
Definition MaxMatch (M: list transaction) (B A: list order):= forall M', Matching M' B A -> Matching M B A -> (Vol M') <= (Vol M).
```

```
Theorem Maximum_Maching
(Process: list order->list order -> instruction ->
(list order)*(list order)*(list transaction))
(B A hat_B hat_A : list order)(tau:instruction):
let B' := (fst (Absorb B A tau)) in
let A' := (snd (Absorb B A tau)) in
let hat_B := (Blist (Process B A tau)) in
let hat_A := (Alist (Process B A tau)) in
let M := (Mlist (Process B A tau)) in
Condition1 M B A hat_B hat_A tau->
Condition3a M B A tau ->
Condition3b M B A hat_B tau ->
Condition3c M B A hat_A tau ->
not (matchable B A) ->
NoDup (ids B') -> NoDup (ids A') ->
MaxMatch (Mlist (Process B A tau)) B' A'.
```

4.2 Local Uniqueness

Theorem 1. Let P_1 and P_2 be processes that satisfy price-time priority, positive bid-ask spread, and conservation. For all order-domain instruction pairs $((B, A), (\Delta, \omega))$ that form legal-inputs, if for each $i \in \{1, 2\}$, $P_i(B, A, (\Delta, \omega)) = (\hat{B_i}, \hat{A_i}, M_i)$, then $(\hat{B_1}, \hat{A_1}, \mathcal{C}(M_1)) = (\hat{B_2}, \hat{A_2}, \mathcal{C}(M_2))$. Furthermore, the following statements hold for each i.

- (1) \hat{B}_i and \hat{A}_i are not matchable.
- (2) The timestamps of orders in \hat{B}_i and \hat{A}_i are distinct and form a subset of the set of timestamps of the orders in $B \cup A \cup \{\mathsf{timestamp}(\omega)\}$.
- (3) The ids of orders in \hat{B}_i and \hat{A}_i are distinct and form a subset of $ids(B \cup A) \cup \{id(\omega)\}.$
- (4) $\Delta = \mathsf{Del} \implies \mathsf{id}(\omega) \notin \mathsf{ids}(\hat{B}_i \cup \hat{A}_i).$

```
Theorem Local_uniqueness
(Process1 Process2: list order->list order -> instruction->
(list order)*(list order)*(list transaction))
(B1 B2 A1 A2 : list order)(tau:instruction):
B1 === B2 -> A1 === A2 -> Legal_input B1 A1 tau ->
Properties Process1 -> Properties Process2 ->
(cform (Mlist (Process1 B1 A1 tau))) ===
(cform (Mlist (Process2 B2 A2 tau)))
/ \setminus
(Blist (Process1 B1 A1 tau)) === (Blist (Process2 B2 A2 tau))
(Alist (Process1 B1 A1 tau)) === (Alist (Process2 B2 A2 tau))
/ \setminus
(*Properties (1) - (4)*)
(included (timesof (Blist (Process1 B1 A1 tau)))
(timesof ((ord tau)::B1)))/\
(included (timesof (Alist (Process1 B1 A1 tau)))
(timesof ((ord tau)::A1)))/\
(not (matchable (Blist (Process1 B1 A1 tau))
(Alist (Process1 B1 A1 tau))))/\
admissible (Blist (Process1 B1 A1 tau))
(Alist (Process1 B1 A1 tau))/\
(included (ids (Blist (Process1 B1 A1 tau)))
(ids ((ord tau)::B1)))/\
(included (ids (Alist (Process1 B1 A1 tau)))
(ids ((ord tau)::A1)))/\
((cmd tau) = del ->
~ In (id (ord tau)) (ids (Blist (Process1 B1 A1 tau))))/\
((cmd tau) = del \rightarrow
~ In (id (ord tau)) (ids (Alist (Process1 B1 A1 tau)))).
```

4.3 Global uniqueness

Theorem 2. Let P_1 and P_2 be processes that satisfy positive bid-ask spread, price-time parity and conservation. Then, for all structured order-books \mathcal{I} and natural numbers k if $\text{Iterated}(P_1, \mathcal{I}, k) = (B_1, A_1, M_1)$ and $\text{Iterated}(P_2, \mathcal{I}, k) = (B_2, A_2, M_2)$, then $(B_1, A_1, C(M_1)) = (B_2, A_2, C(M_2))$.

```
Theorem global_unique (P1 P2 :(list order)->(list order) ->
instruction -> (list order)*(list order)*(list transaction))

(I : list instruction)(k:nat):

(Properties P1) /\((Properties P2) /\) structured I ->

(cform (Mlist (iterated P1 I k))) ===
(cform (Mlist (iterated P2 I k)))

/\

(Blist (iterated P1 I k)) === (Blist (iterated P2 I k))

/\

(Alist (iterated P1 I k)) === (Alist (iterated P2 I k)).
```

4.4 Process_instruction has the desired properties

Theorem 3. Process_instruction satisfies positive bid-ask spread, pricetime priority, and conservation.

```
Theorem Process_correct :
Properties Process_instruction.
```

5 Application: checker

In this section, we discuss how we use our verified algorithm and the global uniqueness theorem to design an automated checker that detects errors in exchange algorithms that run continuous double auctions from their trade-logs.

We implement such a checker and run it on trade-logs from an exchange. We include, as part of the supplementary materials accompanying this paper, the OCaml source code of our checker, trade-logs for two example stocks (created from real stocks from a stock exchange after appropriate pre-processing and masking), and a shell script that compiles our code and runs it on the trade-logs of the two stocks.

As discussed in the paper, exchanges for each traded product maintain an order-book that contains all the incoming orders and a corresponding trade-book that contains all the transactions that are generated as trade-logs. These trade-logs are accessible to the regulators. We show a market regulator can use these trade-logs to automatically determine whether an exchange is complying with the three natural requirements of price-time priority, positive bid-ask spread, and conservation while generating transactions. Recall, as implied by our global uniqueness theorem, these three properties completely specify continuous double auctions.

5.1 Algorithm for checker

Given an order-book and the corresponding trade-book, the checker first runs the verified matching algorithm on the order-book to generate 'verified' matchings. Then, the checker compares the canonical forms of the matchings in the trade-book with the verified matchings, for one time-step at a time. If for any time-step, the canonical forms of the matchings do not match, the checker outputs a mismatch detected message along with the corresponding matchings. Otherwise, if all matchings match, the checker terminates without a mismatch message.

For a given order-book and trade-book, if the above checker finds a mismatch, then we can conclude, from the global uniqueness theorem, that the exchange algorithm violates at least one of the three properties. Otherwise, at least for the instance at hand, the exchange algorithm output is as good as that of a verified algorithm.

In our implementation of the checker, the verified matching algorithm and the canonical form functions are directly extracted from our formalization using Coq's code extraction feature and used as subroutines. To make our extracted code slightly efficient, we mapped the standard arithmetic operators to the corresponding OCaml operators, which ensures our extracted code works using binary arithmetic.

5.2 Pre-processing an order-book

Real exchanges implement more complex instruction types than our three primitives: buy/sell/delete. Here we briefly discuss some of these instruction types and how to convert them into primitive types in a pre-processing step.

 Updates. A buyer or seller might want to update the quantity or price of his order using an update order instruction. For our test exchange the rule is if the quantity of the order decreases, then the timestamp of the original order is retained. Otherwise, if the quantity is increased or the price changes the new order carries the timestamp of the update instruction. Our test exchange as part of the update instruction also maintains the information of quantity of the order that remains untraded. Updates can be easily implemented by replacing it with a delete instruction followed by a new buy/sell instruction with the updated attributes.

- Market orders. Market orders are orders that do not specify a limit price and are ready to be traded at any price. We can implement such an order quite easily by keeping the price 0 for a sell order and ∞ (in our implementation we use the maximum number supported by the system) for a buy order.
- Immediate or cancel orders (IOCs). An IOC is an order that needs to be immediately removed from the system after it is processed, i.e., it should never become a resident order. Such orders can be implemented by replacing it with a buy/sell instruction followed by a delete instruction.
- Stop-loss orders. Stop-loss orders are orders that are triggered when certain events happen, like when the price of a transaction goes below a threshold. For our test exchange, the timestamp of the triggering event is provided. Consequently, these orders can be treated as normal orders when inserted in the order-book at a position corresponding to the timestamp of the triggering event.

Note that the pre-processing step can at the most double the number of instructions in the order-book.

Apart from these standard orders discussed above, certain exchanges allow for more complicated orders which includes 'iceberg orders'. Iceberg orders are rare and are more complicated to pre-process where the priority of the orders depend on factors that cannot be fully determined by just price and time. Although, we can implement such orders through some pre-processing hacks, ideally the matching algorithm should be enriched to handle such orders.

5.3 Demonstration

We create two examples stocks, from two stocks from a real stock exchange after appropriate pre-processing and masking for demonstration purposes.

We collect the order-book and trade-book data of these two stocks for a certain duration of trading, and pre-process the order-book to convert the instructions into primitive instruction types. Also, we mask the ids and timestamps (preserving the ordering) of the orders. We then run our checker on the resulting order-books and the corresponding trade-books.

The checker runs reasonably fast; for the first stock, the order-book has about 16,000 instructions and the checker takes about 7 seconds, and for the second stock, the order-book has about 15,000 instructions and the checker

takes about 6 seconds. For the first stock, no mismatch was detected. For the second stock, a mismatch was detected.

For our exchange we were able to spot similar anomalies as included in the demonstration. We believe that it would be difficult to detect such rare anomalies without the systematic application of formal methods; in absence of formal specifications, uniqueness theorems and verified algorithms, such anomalies can be easily brushed aside as complex and probably unavoidable program behavior.